
1. [bookmark: _Toc122434485][bookmark: _Toc153960358]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 11.7.1 which is part of the 5G NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com
0. [bookmark: _Toc153960359]Verification Test Summary 
Test Case: 	11.7.1
ATS Version:	iwd-TTCN3-B2022-09_D23wk49
System Simulator used:	Anritsu Protocol Conformance Test System ME7834NR
UE used:	Qualcomm Snapdragon X35 5G Modem-RF System UE
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc153960360]Corrections required
Change 2.3 and Change 2.5 of R5s230896 [2] are also required to pass the test case.
1. [bookmark: _Toc153960361]Change 1
	Function name
	function f_TC_11_7_1_NR5GC

	Reason for change
	Current TTCN implementation is not aligned with prose on Table 11.7.1.3.3-1: SIB1 as eDRX-AllowedInactive-r17 is missing

	Summary of change
	Set eDRX-AllowedInactive-r17 to True

	TTCN module
	EDRX_NR5GC.ttcn

	MCC160 Comment
	



Before Change
	  function f_TC_11_7_1_NR5GC() runs on NR5GC_PTC
  {// eDRX / IDLE

    f_NR5GC_Init(NR_1);
  
    //Table 11.7.1.3.3-1: SIB1
    f_NR_CellInfo_Set_SIB1_eDRX_AllowedIdle(nr_Cell1, true_);
    f_NR_CellInfo_SetSIB1_hyperSFN_r17(nr_Cell1, '0000000000'B);

    // Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);

    // UE supporting extended DRX.
    f_UT_SwitchOnUE(UT, false, CNF_REQUIRED);
    f_UT_ConfigureEDRX(UT, true, -, CNF_REQUIRED);
    f_UT_SwitchOffUE(UT, false, CNF_REQUIRED);

    // UE is in state 0N-B according to TS 38.508-1 Table 4.4A.2-2
    f_NR5GC_Preamble (nr_Cell1, STATE_OFF_0B);
    
    f_NR_TestBody_Set(true);
    f_TC_11_7_1_NR5GC_TestBody();
    f_NR_TestBody_Set(false);

    // Switch/Power off UE
    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
  } //end of f_TC_11_7_1_NR5GC



After Change
	  function f_TC_11_7_1_NR5GC() runs on NR5GC_PTC
  {// eDRX / IDLE

    f_NR5GC_Init(NR_1);
  
    //Table 11.7.1.3.3-1: SIB1
    f_NR_CellInfo_Set_SIB1_eDRX_AllowedIdle(nr_Cell1, true_);
    f_NR_CellInfo_SetSIB1_hyperSFN_r17(nr_Cell1, '0000000000'B);
    f_NR_CellInfo_Set_SIB1_eDRX_AllowedInactive(nr_Cell1, true_);

    // Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);

    // UE supporting extended DRX.
    f_UT_SwitchOnUE(UT, false, CNF_REQUIRED);
    f_UT_ConfigureEDRX(UT, true, -, CNF_REQUIRED);
    f_UT_SwitchOffUE(UT, false, CNF_REQUIRED);

    // UE is in state 0N-B according to TS 38.508-1 Table 4.4A.2-2
    f_NR5GC_Preamble (nr_Cell1, STATE_OFF_0B);
    
    f_NR_TestBody_Set(true);
    f_TC_11_7_1_NR5GC_TestBody();
    f_NR_TestBody_Set(false);

    // Switch/Power off UE
    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
  } //end of f_TC_11_7_1_NR5GC


1. [bookmark: _Toc153960362]Change 2
	Function name
	function f_TC_11_7_1_NR5GC_TestBody

	Reason for change
	1. In Step 16-17, current TTCN is not aligned with prose “Steps 19a1-20a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed.”
Also, Step 20a1 RRC Release is missing in current TTCN.
2. In Step 14 and 41, the Extended Paging Time Window has a value of 1 signalled to UE via Registration Accept. It is proposed to use the same parameters for the Paging in Step 18 and Step 47.
3. In Step 26, SRB2 and DRBs are established so signalling shall occur on SRB2.
4. In Step 46, RRC Release is needed prior to Paging in Step 47
5. The test case involves only NR Cell 1. There is a typo in Step 59 that needs to be corrected.
6. In Step 61, UE is being paged so the serviceType in Service Request should be for mobile terminated services.


	Summary of change
	1. Step 16-17 called f_NR5GC_RRC_Idle_Steps16_20 with condition rrcConnectionRelease
2. Change the p_PTWLength_Index in Step 18 and Step 47 from ‘3’ to ‘1’.
3. Step 26 remove input v_SRB to allow Deregistration to happen via SRB2.
4. Step 28-46 add condition rrcConnectionRelease to f_NR5GC_RRC_Idle_Steps16_20
5. Step 59 – fix typo, changing nr_Cell11 to nr_Cell1
6. Step 60-65 – change expected ServiceType to use default value of ‘0010’B.
 

	TTCN module
	EDRX_NR5GC.ttcn

	MCC160 Comment
	



Before Change
	  function f_TC_11_7_1_NR5GC_TestBody() runs on NR5GC_PTC
   { // @sic R5-233411 R5-235344 sic@
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1;
    var template (value) ExtdDRXParams v_NegotiatedExtdDRXParams := cs_NG_NAS_EDRXParams('0000'B, '0101'B,'00000001'B); // Step 14 - Table 11.7.1.3.3-3: REGISTRATION ACCEPT
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NG_NAS_GutiParameters_Type v_GutiParameters := f_NR5GC_CellInfo_GetGutiParameters(nr_Cell1);
    var NG_5G_S_TMSI v_NG_5G_S_TMSI := f_NR_Get5G_S_TMSI_FromGuti(v_GutiParameters);

    //@siclog "Step 1" siclog@
    // Switch on UE.
    f_NR5GC_SwitchOnAndResetIMS();

    //@siclog "Step 2-15" siclog@
    //Steps 2-3 of the generic procedure for NR RRC_IDLE as specified in TS 38.508-1 Table 4.5.2.2-2 are performed.
    //The UE transmits a REGISTRATION REQUEST message including the Requested extended DRX parameters IE
    //Steps 5-13 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 are performed.
    //The SS transmits a REGISTRATION ACCEPT message including the Negotiated extended DRX parameters IE.
    //Check: Does the UE transmit a REGISTRATION COMPLETE message?
    f_NR5GC_RRC_Idle_Steps2_15_eDRX(nr_Cell1, v_NegotiatedExtdDRXParams);
    
    //@siclog "Steps 16-17" siclog@
    // Steps 19a1-20a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 are performed.
    f_NR5GC_PDUSessionEstablishment (nr_Cell1, -, v_SRB);

    //@siclog "Step 18" siclog@
    // The SS transmits a Paging message within a valid paging occasion within the PTW of the next upcoming UE's PH as per Idle eDRX.
    f_NR5GC_UE_Page_eDRX(nr_Cell1,
                         cs_PagingRecordList_OneRecord(cs_PagingUE_Identity_5G_S_TMSI(v_NG_5G_S_TMSI)),
                         3, 5);

    //@siclog "Step 19" siclog@
    // Check: Does the UE transmit an RRCSetupRequest message?
    f_NR_RRC_SetupRequest_Def(nr_Cell1);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.7.1 Step 19");

    //@siclog "Step 20" siclog@
    //The SS transmits an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell1);

    //@siclog "Step 21" siclog@
    // The UE transmit an RRCSetupComplete message including SERVICE REQUEST to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);

    //@siclog "Step 22-25" siclog@
    // Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are executed to successfully complete the service request procedure.
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);

    //@siclog "Step 26" siclog@
    // Switch off UE in RRC CONNECTED as described in TS 38.508-1 subclause 4.9.6.3
    f_NR_UE_DeRegisterOnSwitchOff(nr_Cell1, STATE_CONNECTED_3A, v_SRB);
    
    //@siclog "Step 27" siclog@
    // Switch on UE.
    f_NR5GC_SwitchOnAndResetIMS();
    
    //@siclog "Step 28-46" siclog@
    // The generic test procedure in TS 38.508-1 Table 4.5.2.2-2 indicate that the UE performs registration on NR Cell 1.
    v_NegotiatedExtdDRXParams := cs_NG_NAS_EDRXParams('0000'B, '0001'B, '00000001'B); // Step 31 - Table 11.7.1.3.3-4: REGISTRATION ACCEPT
    f_NR5GC_RRC_Idle_Steps2_15_eDRX(nr_Cell1, v_NegotiatedExtdDRXParams);
    f_NR5GC_RRC_Idle_Steps16_20 (nr_Cell1);
    
    //@siclog "Step 47" siclog@
    // The SS transmits a Paging message in a valid PO.
    f_NR5GC_UE_Page_eDRX(nr_Cell1,
                         cs_PagingRecordList_OneRecord(cs_PagingUE_Identity_5G_S_TMSI(v_NG_5G_S_TMSI)),
                         3, 1);
    
    //@siclog "Step 48" siclog@
    // Check: Does the UE transmit an RRCSetupRequest message?
    f_NR_RRC_SetupRequest_Def(nr_Cell1);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.7.1 Step 48");

    //@siclog "Step 49" siclog@
    //The SS transmits an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell1);

    //@siclog "Step 50" siclog@
    // The UE transmits an RRCSetupComplete message including SERVICE REQUEST to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);

    //@siclog "Step 51-54" siclog@
    // Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are executed to successfully complete the service request procedure.
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
    
    //@siclog "Step 55" siclog@
    // The SS transmits an RRCRelease message to release RRC connection and move the UE to RRC_IDLE.
    f_NR_RRCRelease(nr_Cell1);
    
    //@siclog "Step 56" siclog@
    // The SIB1 is updated according to Table 11.7.1.3.3-1 and transmits a Short message on PDCCH using P-RNTI indicating a systemInfoModification.
    f_NR_CellInfo_SetSIB1_hyperSFN_r17(nr_Cell1, omit); // return value by default
    f_NR_CellInfo_Set_SIB1_eDRX_AllowedIdle(nr_Cell1);
    f_NR_ModifySysinfo(nr_Cell1, true); //Modify Sysinfo in SS
    
    //@siclog "Step 57" siclog@
    //Wait for 2.1* modification period to allow the new system information to take effect.
    f_NR_WaitModificationPeriods (nr_Cell1);
    
    //@siclog "Step 58" siclog@
    // The SS transmits a Paging message to the UE in a valid PO which is derived by normal DRX and not belonging to extended DRX.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Step 59" siclog@
    // Check: Does the UE send an RRCSetupRequest message?
    f_NR_RRC_SetupRequest_Def(nr_Cell11);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.7.1 Step 59");
    
    //@siclog "Step 60-65" siclog@
    // Steps 3-8 of Generic procedure specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell1);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                               tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0000'B)) {
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
    
    
    
   }//end of f_TC_11_7_1_NR5GC_TestBody 



After Change
	  function f_TC_11_7_1_NR5GC_TestBody() runs on NR5GC_PTC
   { // @sic R5-233411 R5-235344 sic@
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1;
    var template (value) ExtdDRXParams v_NegotiatedExtdDRXParams := cs_NG_NAS_EDRXParams('0000'B, '0101'B,'00000001'B); // Step 14 - Table 11.7.1.3.3-3: REGISTRATION ACCEPT
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NG_NAS_GutiParameters_Type v_GutiParameters := f_NR5GC_CellInfo_GetGutiParameters(nr_Cell1);
    var NG_5G_S_TMSI v_NG_5G_S_TMSI := f_NR_Get5G_S_TMSI_FromGuti(v_GutiParameters);

    //@siclog "Step 1" siclog@
    // Switch on UE.
    f_NR5GC_SwitchOnAndResetIMS();

    //@siclog "Step 2-15" siclog@
    //Steps 2-3 of the generic procedure for NR RRC_IDLE as specified in TS 38.508-1 Table 4.5.2.2-2 are performed.
    //The UE transmits a REGISTRATION REQUEST message including the Requested extended DRX parameters IE
    //Steps 5-13 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 are performed.
    //The SS transmits a REGISTRATION ACCEPT message including the Negotiated extended DRX parameters IE.
    //Check: Does the UE transmit a REGISTRATION COMPLETE message?
    f_NR5GC_RRC_Idle_Steps2_15_eDRX(nr_Cell1, v_NegotiatedExtdDRXParams);
    
    //@siclog "Steps 16-17" siclog@
    // Steps 19a1-20a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 are performed.
    f_NR5GC_RRC_Idle_Steps16_20(nr_Cell1, -, -, rrcConnectionRelease);

    //@siclog "Step 18" siclog@
    // The SS transmits a Paging message within a valid paging occasion within the PTW of the next upcoming UE's PH as per Idle eDRX.
    f_NR5GC_UE_Page_eDRX(nr_Cell1,
                         cs_PagingRecordList_OneRecord(cs_PagingUE_Identity_5G_S_TMSI(v_NG_5G_S_TMSI)),
                         1, 5);

    //@siclog "Step 19" siclog@
    // Check: Does the UE transmit an RRCSetupRequest message?
    f_NR_RRC_SetupRequest_Def(nr_Cell1);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.7.1 Step 19");

    //@siclog "Step 20" siclog@
    //The SS transmits an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell1);

    //@siclog "Step 21" siclog@
    // The UE transmit an RRCSetupComplete message including SERVICE REQUEST to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);

    //@siclog "Step 22-25" siclog@
    // Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are executed to successfully complete the service request procedure.
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);

    //@siclog "Step 26" siclog@
    // Switch off UE in RRC CONNECTED as described in TS 38.508-1 subclause 4.9.6.3
    f_NR_UE_DeRegisterOnSwitchOff(nr_Cell1, STATE_CONNECTED_3A);
    
    //@siclog "Step 27" siclog@
    // Switch on UE.
    f_NR5GC_SwitchOnAndResetIMS();
    
    //@siclog "Step 28-46" siclog@
    // The generic test procedure in TS 38.508-1 Table 4.5.2.2-2 indicate that the UE performs registration on NR Cell 1.
    v_NegotiatedExtdDRXParams := cs_NG_NAS_EDRXParams('0000'B, '0001'B, '00000001'B); // Step 31 - Table 11.7.1.3.3-4: REGISTRATION ACCEPT
    f_NR5GC_RRC_Idle_Steps2_15_eDRX(nr_Cell1, v_NegotiatedExtdDRXParams);
    f_NR5GC_RRC_Idle_Steps16_20(nr_Cell1, -, -, rrcConnectionRelease);
    
    //@siclog "Step 47" siclog@
    // The SS transmits a Paging message in a valid PO.
    f_NR5GC_UE_Page_eDRX(nr_Cell1,
                         cs_PagingRecordList_OneRecord(cs_PagingUE_Identity_5G_S_TMSI(v_NG_5G_S_TMSI)),
                         1, 1);
    
    //@siclog "Step 48" siclog@
    // Check: Does the UE transmit an RRCSetupRequest message?
    f_NR_RRC_SetupRequest_Def(nr_Cell1);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.7.1 Step 48");

    //@siclog "Step 49" siclog@
    //The SS transmits an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell1);

    //@siclog "Step 50" siclog@
    // The UE transmits an RRCSetupComplete message including SERVICE REQUEST to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);

    //@siclog "Step 51-54" siclog@
    // Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are executed to successfully complete the service request procedure.
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
    
    //@siclog "Step 55" siclog@
    // The SS transmits an RRCRelease message to release RRC connection and move the UE to RRC_IDLE.
    f_NR_RRCRelease(nr_Cell1);
    
    //@siclog "Step 56" siclog@
    // The SIB1 is updated according to Table 11.7.1.3.3-1 and transmits a Short message on PDCCH using P-RNTI indicating a systemInfoModification.
    f_NR_CellInfo_SetSIB1_hyperSFN_r17(nr_Cell1, omit); // return value by default
    f_NR_CellInfo_Set_SIB1_eDRX_AllowedIdle(nr_Cell1);
    f_NR_ModifySysinfo(nr_Cell1, true); //Modify Sysinfo in SS
    
    //@siclog "Step 57" siclog@
    //Wait for 2.1* modification period to allow the new system information to take effect.
    f_NR_WaitModificationPeriods (nr_Cell1);
    
    //@siclog "Step 58" siclog@
    // The SS transmits a Paging message to the UE in a valid PO which is derived by normal DRX and not belonging to extended DRX.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Step 59" siclog@
    // Check: Does the UE send an RRCSetupRequest message?
    f_NR_RRC_SetupRequest_Def(nr_Cell1);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.7.1 Step 59");
    
    //@siclog "Step 60-65" siclog@
    // Steps 3-8 of Generic procedure specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell1);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                               tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)))) {
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
    
    
    
   }//end of f_TC_11_7_1_NR5GC_TestBody 


[bookmark: _Toc152883446][bookmark: _Toc152883447]
1. [bookmark: _Toc153960363]Change 3
	Function name
	function f_NR5GC_Calculate_PTW_PH

	Reason for change
	Current TTCN computation for UE_ID_H is not in line with spec (TS 38.304 7.4: Paging in extended DRX)

1. Hashed_ID shall be taken only from the bits b31 to b0 of 5G-S-TMSI
2. TeDRX_CN shall be expressed in terms of HSFNs
3. As per TS 38.304, section 7.1, the UE_ID will be 5G-S-TMSI mod 4096 if the UE is operating in eDRX.


	Summary of change
	1. Use substr() to extract the bits b31, b30..., b0 of 5G-S-TMSI
2. Divide p_TeDRX, which is expressed in seconds, by 10.24 to get the HSFN equivalent
3. A new variable v_UE_ID_Mod_Value is defined and initialized with value 4096


	TTCN module
	NR5GC_SysinfoPaging.ttcn

	MCC160 Comment
	



Before Change
	  function f_NR5GC_Calculate_PTW_PH(NR_CellId_Type          p_NR_CellId,
                                        SystemFrameNumber_Type  p_Sfn,
                                        SystemFrameNumber_Type  p_HSFN,
                                        /*hexstring               p_Imsi,*/
                                        float                   p_PTWLength,
                                        float                   p_TeDRX
                                        ) runs on NR5GC_PTC return SubFrameTimingList_Type
      {
        var SubFrameTimingList_Type v_PagingOccasionList;
        var integer v_Index;
        var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (p_NR_CellId);
        var integer v_FirstPagingHSFN;
        var integer v_FirstSfn;
        var SubFrameTiming_Type v_FirstParameters;
        var integer v_IeDRX;//i_s = floor(UE_ID/N) mod Ns
        var integer v_H_UE_ID; // Hashed_ID is Frame Check Sequence (FCS) for the bits b31, b30..., b0 of 5G-S-TMSI
        var NG_5G_S_TMSI v_NG_5G_S_TMSI;
        
        var PagingCycle v_PagingCycle := f_NR_GetSIB1_PagingCycle(p_NR_CellId);
        var PCCH_Config v_PcchConfig := f_NR_GetSIB1_PcchConfig(p_NR_CellId);
        var B32_Type v_NG_TMSI := f_NR5GC_CellInfo_GetNG_TMSI(p_NR_CellId);
        var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
        var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
        var integer v_SlotOffsetPaging := f_NR_GetSlotOffsetForPaging(p_NR_CellId);
      //  var SubFrameTiming_Type v_PagingOccasion;
        
        v_NG_5G_S_TMSI := f_NR_Get5G_S_TMSI_FromGuti(v_GutiParams);
        v_H_UE_ID := bit2int(substr((fx_CalculateFCS32(v_NG_5G_S_TMSI)),0, 13)); // UE_ID_H: 13 most significant bits of the Hashed ID.

        //Calculate next available PH based on v_NextHsfn and HSFN mod T = (UE_ID mod TeDRX)
        v_FirstPagingHSFN := f_NR_Calculate_FirstPagingHSFN (p_Sfn, p_HSFN, float2int(p_TeDRX), v_H_UE_ID);

        //Calculation of PTW values (SFN)
        v_IeDRX := (v_H_UE_ID / float2int(p_TeDRX)) mod 8;
        v_FirstSfn := 128*v_IeDRX; //PTWstart = first radio frame of the PH that is part of the PTW with SFN=128*ieDRX where ieDRX_CN = floor(UE_ID_H /TeDRX_CN) mod 8
        
        
        v_FirstParameters.SFN.Number := v_FirstSfn;// Use this value as only the SFN will be used
        v_FirstParameters.HSFN.Number := v_FirstPagingHSFN;
        
        for (v_Index := 0; v_Index < (float2int(p_PTWLength*100.0)-1); v_Index := v_Index + 1) { //PTWsize = difference between last and first radio frames of the PTW satisfyng SFN formula
          v_PagingOccasionList[v_Index] := f_NR_Calculate_PF_PO(v_FirstParameters,
                                                                v_PagingCycle,
                                                                v_PcchConfig,
                                                                v_NG_TMSI,
                                                                v_SlotsNumberPerSubframe,
                                                                v_SCS,
                                                                v_SlotOffsetPaging);
          //Set the occasion list for each PO fitting on the Paging Time Window
        }

        //Set the occasion list for each PO fitting on the Paging Time Window
        return v_PagingOccasionList;
      }



After Change
	  function f_NR5GC_Calculate_PTW_PH(NR_CellId_Type          p_NR_CellId,
                                        SystemFrameNumber_Type  p_Sfn,
                                        SystemFrameNumber_Type  p_HSFN,
                                        /*hexstring               p_Imsi,*/
                                        float                   p_PTWLength,
                                        float                   p_TeDRX
                                        ) runs on NR5GC_PTC return SubFrameTimingList_Type
      {
        var SubFrameTimingList_Type v_PagingOccasionList;
        var integer v_Index;
        var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (p_NR_CellId);
        var integer v_FirstPagingHSFN;
        var integer v_FirstSfn;
        var SubFrameTiming_Type v_FirstParameters;
        var integer v_IeDRX;//i_s = floor(UE_ID/N) mod Ns
        var integer v_H_UE_ID; // Hashed_ID is Frame Check Sequence (FCS) for the bits b31, b30..., b0 of 5G-S-TMSI
        var NG_5G_S_TMSI v_NG_5G_S_TMSI;
        
        var PagingCycle v_PagingCycle := f_NR_GetSIB1_PagingCycle(p_NR_CellId);
        var PCCH_Config v_PcchConfig := f_NR_GetSIB1_PcchConfig(p_NR_CellId);
        var B32_Type v_NG_TMSI := f_NR5GC_CellInfo_GetNG_TMSI(p_NR_CellId);
        var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
        var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
        var integer v_SlotOffsetPaging := f_NR_GetSlotOffsetForPaging(p_NR_CellId);
        var integer v_UE_ID_Mod_Value := 4096;
      //  var SubFrameTiming_Type v_PagingOccasion;
        
        v_NG_5G_S_TMSI := substr(f_NR_Get5G_S_TMSI_FromGuti(v_GutiParams), 16, 32); // Extract the bits b31, b30..., b0 of 5G-S-TMSI

        v_H_UE_ID := bit2int(substr((fx_CalculateFCS32(v_NG_5G_S_TMSI)),0, 13)); // UE_ID_H: 13 most significant bits of the Hashed ID.

        //Calculate next available PH based on v_NextHsfn and HSFN mod T = (UE_ID mod TeDRX)
        v_FirstPagingHSFN := f_NR_Calculate_FirstPagingHSFN (p_Sfn, p_HSFN, (float2int(p_TeDRX / 10.24)), v_H_UE_ID); // TeDRX in terms of HSFNs

        //Calculation of PTW values (SFN)
        v_IeDRX := (v_H_UE_ID / (float2int(p_TeDRX / 10.24))) mod 8; // TeDRX in terms of HSFNs
        v_FirstSfn := 128*v_IeDRX; //PTWstart = first radio frame of the PH that is part of the PTW with SFN=128*ieDRX where ieDRX_CN = floor(UE_ID_H /TeDRX_CN) mod 8
        
        
        v_FirstParameters.SFN.Number := v_FirstSfn;// Use this value as only the SFN will be used
        v_FirstParameters.HSFN.Number := v_FirstPagingHSFN;
        
        for (v_Index := 0; v_Index < (float2int(p_PTWLength*100.0)-1); v_Index := v_Index + 1) { //PTWsize = difference between last and first radio frames of the PTW satisfyng SFN formula
          v_PagingOccasionList[v_Index] := f_NR_Calculate_PF_PO(v_FirstParameters,
                                                                v_PagingCycle,
                                                                v_PcchConfig,
                                                                v_NG_TMSI,
                                                                v_SlotsNumberPerSubframe,
                                                                v_SCS,
                                                                v_SlotOffsetPaging,
                                                                v_UE_ID_Mod_Value);
          //Set the occasion list for each PO fitting on the Paging Time Window
        }

        //Set the occasion list for each PO fitting on the Paging Time Window
        return v_PagingOccasionList;
      }



1. [bookmark: _Toc153869091][bookmark: _Toc153960364]Change 4
	[bookmark: _Hlk153874708]Function name
	f_NR_Calculate_PF_PO

	Reason for change
	1. As per TS 38.304, section 7.1, the UE_ID will be 5G-S-TMSI mod 4096 if the UE is operating in eDRX.
2. As per TS 38.304 section 7.1, if the UE operates in eDRX mode, then the Paging cycle TeDRX is the eDRX cycle, and not the paging cycle value from SIB1

	Summary of change
	1. Passed a new argument p_UE_ID_Mod_Value  in function f_NR_Calculate_PF_PO() for the mod value to calculate the UE_ID to handle the case when UE operates in eDRX mode.
2. Passed a new argument p_eDRX_Value  in function f_NR_Calculate_PF_PO() for eDRX value to handle the cases where the UE is operating in eDRX mode, and used this value to calculate the v_T (TeDRX) and consequently v_N (total number of Paging frames in T)

	TTCN module
	NR5GC_SysinfoPaging.ttcn

	MCC160 Comment
	



Before Change:
	   function f_NR_Calculate_PF_PO (SubFrameTiming_Type    p_CurrentTiming,
                                 PagingCycle             p_T,
                                 PCCH_Config             p_PCCH_Config,
                                 B32_Type                p_NG_TMSI,
                                 integer                 p_SlotsNumberPerSubframe,
                                 SubcarrierSpacing       p_SCS,
                                 integer                 p_SlotOffset) return SubFrameTiming_Type
  {
    //@sic R5s190291 change 1: change first parameter p_Sfn -> p_CurrentTiming (type SubFrameTiming_Type) sic@
    var integer v_UE_ID := bit2int(p_NG_TMSI) mod 1024;                         // 5G-S-TMSI mod 1024
    var integer v_PF_Offset := fl_NR_ConvertPF_Offset(p_PCCH_Config.nAndPagingFrameOffset);
    // var integer v_Ns := fl_NR_ConvertNs(p_PCCH_Config.ns);                 //number of paging occasions for a PF
    var integer v_NextSfn := p_CurrentTiming.SFN.Number + tsc_NR_SfnDelay; //Absolute value - no mod 1024
    var integer v_SfnPlusOffsetModT;
    var integer v_FirstPagingFrame;
    var integer v_T := fl_NR_ConvertPagingCycle(p_T);
    var integer v_N := fl_NR_GetN(p_PCCH_Config.nAndPagingFrameOffset, v_T);     // N: number of total paging frames in T @sic R5s190150 change 4 sic@
    //   var integer v_I_s;
    var SubFrameTiming_Type v_PagingOccasion;
    
    //Calculate the first paging Frame
    //SFN for the PF is determined by: (SFN + PF_offset) mod T = (T div N)*(UE_ID mod N)
    //Absolute value - no mod 1024 @sic R5-211417 sic@
    v_SfnPlusOffsetModT := (v_T / v_N) * (v_UE_ID mod v_N);
    if ((v_NextSfn mod v_T) < v_SfnPlusOffsetModT) {
      v_FirstPagingFrame := ((v_NextSfn - v_NextSfn mod v_T) + v_SfnPlusOffsetModT) - v_PF_Offset;
    } else {
      v_FirstPagingFrame := (v_NextSfn - v_NextSfn mod v_T) + v_SfnPlusOffsetModT + v_T - v_PF_Offset;
    }
   //* Calculate i_s = floor (UE_ID/N) mod Ns
    // v_I_s := (v_UE_ID mod v_N) mod v_Ns; Note: unused so far
    //PDCCH monitoring occasions according to paging-SearchSpace and firstPDCCH-MonitoringOccasionOfPO
    //PDCCH monitoring occasions according to the default association: PDCCH monitoring occasions for paging are same as for RMSI as defined in Section 13 in TS 38.213
    //For Ns = 1, there is only one PO which starts in the PF.
    //For Ns = 2, PO is either in the first half frame (i_s = 0) or the second half frame (i_s = 1) of the PF.
    v_PagingOccasion.HSFN.Number := (p_CurrentTiming.HSFN.Number + (v_FirstPagingFrame)/1024 ) mod 1024;  //@sic R5-211417 sic@
    v_PagingOccasion.SFN.Number := v_FirstPagingFrame mod 1024;                                           //@sic R5-211417 sic@
    v_PagingOccasion.Subframe.Number := p_SlotOffset / p_SlotsNumberPerSubframe;
    v_PagingOccasion.Symbol := valueof(cs_SymbolTimingInfo_FirstSlot); //@sic R5s221246 sic@
    v_PagingOccasion := f_NR_SubFrameTiming_SetSlot(v_PagingOccasion,
                                                    p_SlotOffset mod p_SlotsNumberPerSubframe,
                                                    p_SCS);
    return v_PagingOccasion;
  }  





After Change:
	   function f_NR_Calculate_PF_PO (SubFrameTiming_Type    p_CurrentTiming,
                                 PagingCycle             p_T,
                                 PCCH_Config             p_PCCH_Config,
                                 B32_Type                p_NG_TMSI,
                                 integer                 p_SlotsNumberPerSubframe,
                                 SubcarrierSpacing       p_SCS,
                                 integer                 p_SlotOffset,
                                 integer                 p_UE_ID_Mod_Value := 1024,
                                 template (omit) float          p_eDRX_Value := omit) return SubFrameTiming_Type
  {
    //@sic R5s190291 change 1: change first parameter p_Sfn -> p_CurrentTiming (type SubFrameTiming_Type) sic@
    var integer v_UE_ID := bit2int(p_NG_TMSI) mod p_UE_ID_Mod_Value;                         // 5G-S-TMSI mod 1024 or 4096
    var integer v_PF_Offset := fl_NR_ConvertPF_Offset(p_PCCH_Config.nAndPagingFrameOffset);
    //   var integer v_Ns := fl_NR_ConvertNs(p_PCCH_Config.ns);                 //number of paging occasions for a PF
    var integer v_NextSfn := p_CurrentTiming.SFN.Number + tsc_NR_SfnDelay; //Absolute value - no mod 1024
    var integer v_SfnPlusOffsetModT;
    var integer v_FirstPagingFrame;
    var integer v_T := fl_NR_ConvertPagingCycle(p_T);
    var integer v_N; // N: number of total paging frames in T @sic R5s190150 change 4 sic@
    //   var integer v_I_s;
    var SubFrameTiming_Type v_PagingOccasion;
    
    if(ispresent(p_eDRX_Value))
    {
      v_T := float2int(p_eDRX_Value * 100.00);
    }
    
    v_N := fl_NR_GetN(p_PCCH_Config.nAndPagingFrameOffset, v_T); 

    //Calculate the first paging Frame
    //SFN for the PF is determined by: (SFN + PF_offset) mod T = (T div N)*(UE_ID mod N)
    //Absolute value - no mod 1024 @sic R5-211417 sic@
    v_SfnPlusOffsetModT := (v_T / v_N) * (v_UE_ID mod v_N);
    if ((v_NextSfn mod v_T) < v_SfnPlusOffsetModT) {
      v_FirstPagingFrame := ((v_NextSfn - v_NextSfn mod v_T) + v_SfnPlusOffsetModT) - v_PF_Offset;
    } else {
      v_FirstPagingFrame := (v_NextSfn - v_NextSfn mod v_T) + v_SfnPlusOffsetModT + v_T - v_PF_Offset;
    }
   //* Calculate i_s = floor (UE_ID/N) mod Ns
    // v_I_s := (v_UE_ID mod v_N) mod v_Ns; Note: unused so far
    //PDCCH monitoring occasions according to paging-SearchSpace and firstPDCCH-MonitoringOccasionOfPO
    //PDCCH monitoring occasions according to the default association: PDCCH monitoring occasions for paging are same as for RMSI as defined in Section 13 in TS 38.213
    //For Ns = 1, there is only one PO which starts in the PF.
    //For Ns = 2, PO is either in the first half frame (i_s = 0) or the second half frame (i_s = 1) of the PF.
    v_PagingOccasion.HSFN.Number := (p_CurrentTiming.HSFN.Number + (v_FirstPagingFrame)/1024 ) mod 1024;  //@sic R5-211417 sic@
    v_PagingOccasion.SFN.Number := v_FirstPagingFrame mod 1024;                                           //@sic R5-211417 sic@
    v_PagingOccasion.Subframe.Number := p_SlotOffset / p_SlotsNumberPerSubframe;
    v_PagingOccasion.Symbol := valueof(cs_SymbolTimingInfo_FirstSlot); //@sic R5s221246 sic@
    v_PagingOccasion := f_NR_SubFrameTiming_SetSlot(v_PagingOccasion,
                                                    p_SlotOffset mod p_SlotsNumberPerSubframe,
                                                    p_SCS);
    return v_PagingOccasion;
  }


1. [bookmark: _Toc153869092][bookmark: _Toc153960365]Change 5
	Function name
	f_NR5GC_PO_List_eDRX

	Reason for change
	1. As per TS 38.304, section 7.1, the UE_ID will be 5G-S-TMSI mod 4096 if the UE is operating in eDRX.
2. eDRX value is required to calculate the TeDRX  in function f_NR_Calculate_PF_PO()

	Summary of change
	1. Added a new variable v_UE_ID_Mod_Value and initialized with value 4096, as this function is used when the UE operates in eDRX mode
2. Passed the value of eDRX as an argument to function f_NR_Calculate_PF_PO()

	TTCN module
	NR5GC_SysinfoPaging.ttcn

	MCC160 Comment
	



Before Change:
	        function f_NR5GC_PO_List_eDRX(NR_CellId_Type                 p_NR_CellId,
                                    integer                        p_PTWLength_Index,
                                    integer                        p_EDRXValue_Index) runs on NR5GC_PTC return SubFrameTimingList_Type
      {
        var SubFrameTiming_Type v_CurrentTiming;
        var PCCH_Config v_PcchConfig := f_NR_GetSIB1_PcchConfig(p_NR_CellId);
        var B32_Type v_NG_TMSI := f_NR5GC_CellInfo_GetNG_TMSI(p_NR_CellId);
        var SubFrameTimingList_Type v_PoList := {};
        var SubFrameTiming_Type v_PagingOccasion;
        var float v_EDRXValue;
        var float v_PTWLengthValue;
        var PagingCycle v_PagingCycle := f_NR_GetSIB1_PagingCycle(p_NR_CellId);
            
        var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
        var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
        var integer v_SlotOffsetPaging := f_NR_GetSlotOffsetForPaging(p_NR_CellId);
        
        v_PTWLengthValue := tsc_PTWLengthParameters[p_PTWLength_Index];
        v_EDRXValue := tsc_NR5GC_eDRX_Cycle_Length[p_EDRXValue_Index];
        
        //Get current SFN and subframe
        v_CurrentTiming := f_NR_GetCurrentTiming(p_NR_CellId);

        // Calculate next Paging Occasion
        // TS 38.304 clause 7.4: If the UE is configured with an extended DRX cycle no longer than 1024 radio frames, it monitors POs as defined in 7.1 with configured eDRX cycle. Otherwise, a UE configured with eDRX monitors POs as defined in 7.1 during a periodic Paging Time Window (PTW) configured for the UE.
        if (p_EDRXValue_Index < 3){ // TS 38.304 cl. 7.1: calculate the first Paging occasion
          v_PagingOccasion := f_NR_Calculate_PF_PO(v_CurrentTiming,
                                                   v_PagingCycle,
                                                   v_PcchConfig,
                                                   v_NG_TMSI,
                                                   v_SlotsNumberPerSubframe,
                                                   v_SCS,
                                                   v_SlotOffsetPaging);
          v_PoList[0] := v_PagingOccasion;
        }
        else {
          v_PoList := f_NR5GC_Calculate_PTW_PH(p_NR_CellId, v_CurrentTiming.SFN.Number, v_CurrentTiming.HSFN.Number, v_PTWLengthValue, v_EDRXValue);
        }
     
        return v_PoList;
      } 





After Change:
	      function f_NR5GC_PO_List_eDRX(NR_CellId_Type                 p_NR_CellId,
                                    integer                        p_PTWLength_Index,
                                    integer                        p_EDRXValue_Index) runs on NR5GC_PTC return SubFrameTimingList_Type
      {
        var SubFrameTiming_Type v_CurrentTiming;
        var PCCH_Config v_PcchConfig := f_NR_GetSIB1_PcchConfig(p_NR_CellId);
        var B32_Type v_NG_TMSI := f_NR5GC_CellInfo_GetNG_TMSI(p_NR_CellId);
        var SubFrameTimingList_Type v_PoList := {};
        var SubFrameTiming_Type v_PagingOccasion;
        var float v_EDRXValue;
        var float v_PTWLengthValue;
        var PagingCycle v_PagingCycle := f_NR_GetSIB1_PagingCycle(p_NR_CellId);
            
        var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
        var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
        var integer v_SlotOffsetPaging := f_NR_GetSlotOffsetForPaging(p_NR_CellId);
        var integer v_UE_ID_Mod_Value := 4096;
        
        v_PTWLengthValue := tsc_PTWLengthParameters[p_PTWLength_Index];
        v_EDRXValue := tsc_NR5GC_eDRX_Cycle_Length[p_EDRXValue_Index];
        
        //Get current SFN and subframe
        v_CurrentTiming := f_NR_GetCurrentTiming(p_NR_CellId);

        // Calculate next Paging Occasion
        // TS 38.304 clause 7.4: If the UE is configured with an extended DRX cycle no longer than 1024 radio frames, it monitors POs as defined in 7.1 with configured eDRX cycle. Otherwise, a UE configured with eDRX monitors POs as defined in 7.1 during a periodic Paging Time Window (PTW) configured for the UE.
        if (p_EDRXValue_Index < 3){ // TS 38.304 cl. 7.1: calculate the first Paging occasion
          v_PagingOccasion := f_NR_Calculate_PF_PO(v_CurrentTiming,
                                                   v_PagingCycle,
                                                   v_PcchConfig,
                                                   v_NG_TMSI,
                                                   v_SlotsNumberPerSubframe,
                                                   v_SCS,
                                                   v_SlotOffsetPaging,
                                                   v_UE_ID_Mod_Value,
                                                   v_EDRXValue);
          v_PoList[0] := v_PagingOccasion;
        }
        else {
          v_PoList := f_NR5GC_Calculate_PTW_PH(p_NR_CellId, v_CurrentTiming.SFN.Number, v_CurrentTiming.HSFN.Number, v_PTWLengthValue, v_EDRXValue);
        }
     
        return v_PoList;
      }




1. [bookmark: _Toc121172016][bookmark: _Toc151982517][bookmark: _Toc153960366]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on Anritsu with NR band n41 using NR ciphering nea2 and integrity algorithm nia2. The test case was executed on Keysight with NR band n78 using ciphering algorithm nea1 and integrity algorithm nia1.
3 [bookmark: _Toc121172017][bookmark: _Toc151982518][bookmark: _Toc153960367]Execution Log Files
3.1 [bookmark: _Toc151982519][bookmark: _Toc153960368][bookmark: _Hlk120026754]Qualcomm Snapdragon x35 5G Modem-RF System UE
The Qualcomm Snapdragon x35 5G Modem RF System UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR and Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Anritsu/TC_11_7_1_LOG.xml
Keysight/TC_11_7_1_LOG.html

· PICS/PIXIT parameter file:
Anritsu/TC_11_7_1_PIXIT.txt
Keysight/TC_11_7_1_PIXIT.xml

4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc121172019][bookmark: _Toc151982520][bookmark: _Toc153960369]References 
	[1]
	R5s240062: Supporting information for addition of NR5GC R17 eDRX testcase 11.7.1 in FR1

	[2]
	R5s230896: Addition of NR5GC testcase 11.7.2 in FR1



